# Using *Mathematica* to Teach Linear Differential Operators and the Method of Undetermined Coefficients

Itai Seggev
Knox College / Wolfram Research

Joint Mathematics Meetings

January 9, 2011

New Orleans, Louisiana

## Problem Statement

Given a linear differential operator $P(D)$ and a polynomial-exponential forcing function $f$, we wish to find

1. the general solution to the homogeneous equation $P(D)[y] = 0$
2. one solution to the driven equation $P(D)[y] = f$ by means of an ansatz

We can then solve any IVP for $P(D)$, but

3. we need to enter $P(D)$ into *Mathematica*

## Ingredients

4. Solve[]

5. CoefficientList[]

6. ComplexExpand[], Re[], Im[] (if dealing with complex and or trigonometric equations)

7. Some magic code:

In[11]:=
```
Clear[DPlusS]
DPlusS /: Power[DPlusS[t_, s_:0], n_] := Function[f,Nest[(D[#, t] + s #)&, f, n]]
DPlusS[t_, s_:0]^n_[f___] ^:= DPlusS[t, s]^n[f]
DPlusS[t_,s_:0][f_]:=D[f,t]+s f
Unprotect[Plus, Times];
Plus  /: Plus[ (a_:1) o1_Function , cruft__][f_] := a  o1[f] + Plus[cruft][f]
Plus  /: Plus[ (a_:1)DPlusS[t_, s___] , cruft__][f_] := a  DPlusS[t, s][f] + Plus[cruft][f]
Times /: Times[a_, DPlusS[t_,s___] ][f_] := a DPlusS[t,s][f]
Times /: Times[a_, o1_Function ][f_] := a o1[f]
Protect[Plus, Times];
```

# Playing with DPlusS[$t$, $s$] = $D + s$

## The basic operator

In[21]:=
```
DPlusS[t, s][y[t]]
```

Out[21]= $s\, y[t] + y'[t]$

## Monomials

Positive powers:

In[22]:=
```
DPlusS[t, 0]^2[y[t]]
DPlusS[t]^2[y[t]]
```

Out[22]= $y''[t]$

Out[23]= $y''[t]$

Positive powers with shift:

In[24]:=
```
DPlusS[t, s]^2[y[t]] // Expand
```

Out[24]= $s^2\, y[t] + 2\, s\, y'[t] + y''[t]$

The identity operator

In[25]:=
```
DPlusS[t, s]^0[y[t]]
```

Out[25]= $y[t]$

## A polynomial

In[26]:=
```
Clear[PofD]
```

$$\text{PofD}[t\_, s\_ : 0] := L\, \text{DPlusS}[t, s]\text{^}2 + R\, \text{DPlusS}[t, s] + \frac{\text{DPlusS}[t, s]^0}{C}$$

```
PofD[t, 0][y[t]]
```

Out[28]= $\frac{y[t]}{C} + R\, y'[t] + L\, y''[t]$

In[29]:=
```
PofD[t, ⅈ ω][y[t]] // Simplify
```

Out[29]= $\left(\frac{1}{C} + \omega\, (i\, R - L\, \omega)\right) y[t] + (R + 2\, i\, L\, \omega)\, y'[t] + L\, y''[t]$

# Step 1: Finding the basic solution set

## Simple roots

Assuming a constant coefficient operator, we know the solution is given by the characteristic roots

In[30]:=
```
PofD[t_, s_ : 0] := DPlusS[t, s]^2 + 5 DPlusS[t, s] + 6 DPlusS[t, s]^0
```

The characteristic polynomial is obtain with $D \to 0$, i.e., computing $P(D)[1]$

In[31]:=
```
PofD[t, s][1]
Solve[% == 0, s]
Exp[s t] /. %
```

Out[31]= $6 + 5 s + s^2$

Out[32]= {{s → -3}, {s → -2}}

Out[33]= $\{e^{-3 t}, e^{-2 t}\}$

So, we have our basic solution set.

## Repeated roots

In[34]:=
```
PofD[t_, s_ : 0] := DPlusS[t, s]^2 + 4 DPlusS[t, s] + 4 DPlusS[t, s]^0
PofD[t, s][1]
Solve[% == 0, s]
Exp[s t] /. %
```

Out[35]= $4 + 4 s + s^2$

Out[36]= {{s → -2}, {s → -2}}

Out[37]= $\{e^{-2 t}, e^{-2 t}\}$

Our set is linearly dependent, we need to replace one vector with $t\, e^{-2 t}$

## Step 2: Find the particular solution (real case)

In[38]:=
```
Clear[PofD]
PofD[t_, s_ : 0] := DPlusS[t, s]^2 + 5 DPlusS[t, s] + 6 DPlusS[t, s]^0
```

A couple of forcing functions

In[40]:=
```
f1[t_] := t^2 + 3
f2[t_] := Exp[7 t] (t + 2)
```

### ▪ $f_1$

We know that the particular solution has the same form as the driving polynomial in the case of a simple polynomial driving term.

In[42]:=
```
y1[t_] := a2 t^2 + a1 t + a0
```

We must substitute in and solve for the coefficients. CoefficientList[] to the rescue!

In[43]:=
```
CoefficientList[f1[t], t]
CoefficientList[PofD[t][y1[t]], t]
Solve[% == %%, {a0, a1, a2}]
y1[t] /. First[%]
```

Out[43]= {3, 0, 1}

Out[44]= {6 a0 + 5 a1 + 2 a2, 6 a1 + 10 a2, 6 a2}

Out[45]= $\left\{\left\{a0 \to \frac{73}{108}, a1 \to -\frac{5}{18}, a2 \to \frac{1}{6}\right\}\right\}$

Out[46]= $\frac{73}{108} - \frac{5\,t}{18} + \frac{t^2}{6}$

### ▪ $f_2$

When there is an exponential term, it is conventional to separate the solution into its polynomial and exponential parts.

In[47]:=
```
Clear[y2, h2]
h2[t_] := a1 t + a0
y2[t_] := h2[t] Exp[7 t]
```

We then equate the polynomial part of the driving terms with the shifted operator acting on the polynomial ansatz:

```
In[50]:= CoefficientList[f2[t] / Exp[7 t], t]
         CoefficientList[PofD[t, 7][h2[t]], t]
         Solve[% == %%, {a0, a1}]
         y2[t] /. First[%]
```

Out[50]= $\{2, 1\}$

Out[51]= $\{90\, a0 + 19\, a1, 90\, a1\}$

Out[52]= $\left\{\left\{a0 \to \frac{161}{8100}, a1 \to \frac{1}{90}\right\}\right\}$

Out[53]= $e^{7\, t} \left(\frac{161}{8100} + \frac{t}{90}\right)$

- **Alternate solution**

While it is conventional to do the separation, it is not strictly required.   CoefficientList[] will consider the exponential part of the "coefficient", allowing its cancellation.

```
In[54]:= CoefficientList[f2[t], t]
         CoefficientList[PofD[t][y2[t]], t]
         Solve[% == %%, {a0, a1}]
         y2[t] /. First[%]
```

Out[54]= $\{2\, e^{7\, t}, e^{7\, t}\}$

Out[55]= $\{90\, a0\, e^{7\, t} + 19\, a1\, e^{7\, t}, 90\, a1\, e^{7\, t}\}$

Out[56]= $\left\{\left\{a0 \to \frac{161}{8100}, a1 \to \frac{1}{90}\right\}\right\}$

Out[57]= $e^{7\, t} \left(\frac{161}{8100} + \frac{t}{90}\right)$

## Step 2: Find the particular solution (trig/complex cases)

### ■ Sinusoidal driving term

In[58]:=
```
Clear[PofD, f]
PofD[t_, s_ : 0] := DPlusS[t, s]^2 + 5 DPlusS[t, s] + 6 DPlusS[t, s]^0
f[t_] := Sin[7 t] (t + 2)
```

Since it we have a sinusoid, we need to complexify the problem, solving with a driving term $f_c(t) = (t+2)\, e^{7\,i\,t}$.

In[61]:=
```
Clear[yc, hc]
hc[t_] := a1 t + a0
fc[t_] :=  Exp[7 ⅈ t] (t + 2)
yc[t_] := hc[t] Exp[7 ⅈ t]
```

Finding the complex solution is as before.

In[65]:=
```
CoefficientList[fc[t] / Exp[7 ⅈ t], t]
CoefficientList[PofD[t, 7 ⅈ][hc[t]], t]
Solve[% == %%, {a0, a1}]
yc[t_] = yc[t] /. First[%]
```

Out[65]= {2, 1}

Out[66]= {(-43 + 35 ⅈ) a0 + (5 + 14 ⅈ) a1, (-43 + 35 ⅈ) a1}

Out[67]= $\left\{\left\{a0 \to -\frac{56\,336}{2\,362\,369} - \frac{119\,483\, i}{4\,724\,738},\ a1 \to -\frac{43}{3074} - \frac{35\, i}{3074}\right\}\right\}$

Out[68]= $e^{7\,i\,t}\left(\left(-\frac{56\,336}{2\,362\,369} - \frac{119\,483\, i}{4\,724\,738}\right) - \left(\frac{43}{3074} + \frac{35\, i}{3074}\right) t\right)$

We can then find the solution by taking the imaginary part of our complex solution.

In[69]:=
```
y[t_] = ComplexExpand[Im[yc[t]]] // Simplify
```

Out[69]= $\frac{-7\,(17\,069 + 7685\, t)\, \mathrm{Cos}[7\, t] - (112\,672 + 66\,091\, t)\, \mathrm{Sin}[7\, t]}{4\,724\,738}$

Notice that the ComplexExpand[] is essential, since the value above is not valid if $t$ is complex.

In[70]:=
```
Im[yc[t]] // Simplify
```

Out[70]= $\mathrm{Im}\left[\left(-\frac{1505}{4\,724\,738} + \frac{156\, i}{2\,362\,369}\right) e^{7\,i\,t}\,((56 + 91\, i) + (35 + 43\, i)\, t)\right]$

For free, we also get the solution for a cosine driving term.

In[71]:=
```
ComplexExpand[Re[yc[t]]] // Simplify
```

Out[71]= $\frac{-(112\,672 + 66\,091\, t)\, \mathrm{Cos}[7\, t] + 7\,(17\,069 + 7685\, t)\, \mathrm{Sin}[7\, t]}{4\,724\,738}$

### ■ Alternate solution

As before, we are not obligated to separate out the polynomial if we don't want to.

In[72]:=
```
yc[t_] := hc[t] Exp[7 ⅈ t]
CoefficientList[fc[t], t]
CoefficientList[PofD[t][yc[t]], t]
Solve[% == %%, {a0, a1}]
```

Out[73]= $\{2\, e^{7 i t},\ e^{7 i t}\}$

Out[74]= $\{(-43 + 35\, i)\, a0\, e^{7 i t} + (5 + 14\, i)\, a1\, e^{7 i t},\ (-43 + 35\, i)\, a1\, e^{7 i t}\}$

Out[75]= $\left\{\left\{a0 \to -\frac{56336}{2362369} - \frac{119483\, i}{4724738},\ a1 \to -\frac{43}{3074} - \frac{35\, i}{3074}\right\}\right\}$

## Bonus slide: additional cases

### Degeneracy

A degeneracy occurs when the characteristic polynomial has a repeated root, and/or the shift equals one of the roots.

In[76]:=
```
Clear[PofD, f]
PofD[t_, s_ : 0] := DPlusS[t, s]^2 + 4 DPlusS[t, s] + 4 DPlusS[t, s]^0
f[t_] := (t^2 + 2 t + 3) Exp[-2 t]
```

In[79]:=
```
Clear[y, h]
h[t_] := a2 t^2 + a1 t + a0
y[t_] := h[t] Exp[-2 t]
```

Now, what happens if we proceed with our normal ansatz without paying attention to the degeneracy?

In[82]:=
```
CoefficientList[f[t] / Exp[-2 t], t]
CoefficientList[PofD[t, -2][h[t]], t]
Solve[% == %%, {a0, a1, a2}]
```

Out[82]= {3, 2, 1}

Out[83]= {2 a2}

Out[84]= {}

We get no solution: something has gone wrong. But, there is more information. The two lowest terms have dropped out entirely: we have a double degeneracy (repeated root equal to shift)

Hence, we use a polynomial of degree two higher (but possessing the same number of terms).

In[85]:=
```
h[t_] := a4 t^4 + a3 t^3 + a2 t^2
CoefficientList[f[t] / Exp[-2 t], t]
CoefficientList[PofD[t, -2][h[t]], t]
Solve[% == %%, {a2, a3, a4}]
y[t] /. First[%]
```

Out[86]= {3, 2, 1}

Out[87]= {2 a2, 6 a3, 12 a4}

Out[88]= $\left\{\left\{a2 \to \frac{3}{2},\ a3 \to \frac{1}{3},\ a4 \to \frac{1}{12}\right\}\right\}$

Out[89]= $e^{-2t}\left(\frac{3t^2}{2} + \frac{t^3}{3} + \frac{t^4}{12}\right)$

### Higher order operators

In[90]:=
```
Clear[PofD, f]
PofD[t_, s_ : 0] :=
 DPlusS[t, s]^4 + 4 DPlusS[t, s]^3 + 3 DPlusS[t, s]^2 - 4 DPlusS[t, s] - 4 DPlusS[t, s]^0
f[t_] := (t^2 + 2 t + 3) Exp[-t]
```

Characteristic roots are -2, -2, 1, and -1.

In[93]:=
```
PofD[t, s][1] // Simplify
```
Out[93]= $(2 + s)^2 \left(-1 + s^2\right)$

We have single degenerac: shift equal to a simple root. So we need a polynomial of one degree higher. Rest of process is unchanged.

In[94]:=
```
Clear[y, h]
h[t_] := a3 t^3 + a2 t^2 + a1 t
y[t_] := h[t] Exp[-t]
CoefficientList[f[t] / Exp[-t], t]
CoefficientList[PofD[t, -1][h[t]], t]
Solve[% == %%, {a1, a2, a3}]
y[t] /. First[%]
```
Out[97]= {3, 2, 1}

Out[98]= {-2 a1 - 6 a2, -4 a2 - 18 a3, -6 a3}

Out[99]= $\left\{\left\{a1 \to -\frac{9}{4},\ a2 \to \frac{1}{4},\ a3 \to -\frac{1}{6}\right\}\right\}$

Out[100]= $e^{-t} \left(-\frac{9\,t}{4} + \frac{t^2}{4} - \frac{t^3}{6}\right)$

## ■ RLC circuit (complete example from lecture notes)

An RLC circuit obeys

$$L\,\ddot{q} + R\,\dot{q} + \frac{q}{C} == \mathcal{E}(t) \tag{1}$$

If we wanted to express the relationship in terms of current, we would need to differentiate the previous equation to get:

$$L\,\ddot{I} + R\,\dot{I} + \frac{I}{C} == \dot{\mathcal{E}}(t). \tag{2}$$

Getting equation (2) is helpful because expressing charge in terms of current involves a definite integral: $q(t) = q(t_0) + \int_{t_0}^{t} I(s)\,ds$.

If we substitute this equation into equation (1), we'd get a so-called integro-differential equation. Differentiating both sides with respect to $t$ gives us a plain old differential equation.

For convenience, we put a bunch of assumptions into the global variable **$Assumptions** which will shorten many **Simplify[]** commands.

In[101]:=
```
$Assumptions = {a ∈ Reals, R > 0, ω > 0, L > 0, C > 0, t ∈ Reals}
```
Out[101]= {a ∈ Reals, R > 0, ω > 0, L > 0, C > 0, t ∈ Reals}

We define our differential operator

In[102]:=
```
Clear[PofD]
PofD[t_, s_ : 0] := L DPlusS[t, s]^2 + R DPlusS[t, s] + DPlusS[t, s]^0 / C
```

We want to solve equation (2) assuming $\dot{\mathcal{E}}(t) = a\cos(\omega t)$. First, we must find the characteristic roots:

In[104]:=
```
Solve[PofD[t, s][1] == 0, s] // Simplify
```

Out[104]= $\left\{\left\{s \to -\frac{R + \sqrt{-\frac{4L}{C} + R^2}}{2L}\right\}, \left\{s \to \frac{-R + \sqrt{-\frac{4L}{C} + R^2}}{2L}\right\}\right\}$

As long as $R$ is non-zero (i.e., there actually is a resistor in the circuit), the characteristic roots have non-zero real part and hence cannot equal $s = i\omega$. This eliminates degenerate cases of undetermined coefficients. We use *Mathematica* to find the solution, but as in example 3 this is unnecessary. Since the polynomial multiplying the cosine is a constant, the polynomial $h(t)$ in the driven solution is

$$h(t) = \frac{a}{P(i\,\omega)}$$

We ask *Mathematica* to verify:

In[105]:=
```
Clear[h, y, z]
h[t_] := a0
Solve[CoefficientList[PofD[t, ⅈ ω][h[t]], t] == {a}, a0]
z[t_] = h[t] ⅇ^(ⅈ ω t) /. %[[1]]
```

Out[107]= $\left\{\left\{a0 \to \frac{a}{\frac{1}{C} + i R \omega - L \omega^2}\right\}\right\}$

Out[108]= $\frac{a\, e^{i t \omega}}{\frac{1}{C} + i R \omega - L \omega^2}$

Since we're focused on a cosine driving term, we need to take the real part of $z$. The following would be the steps corresponding to finding the real part by hand.

We make the denominator real by multiplying by its complex conjugate:

In[109]:= $\left(\frac{1}{C} + i R \omega - L \omega^2\right)\left(\frac{1}{C} - i R \omega - L \omega^2\right)$ `// Simplify`

Out[109]= $\frac{1}{C^2} - \frac{2 L \omega^2}{C} + R^2 \omega^2 + L^2 \omega^4$

We do the same thing to the numerator and use Euler's identity to convert to sines and cosines.

In[110]:= `ExpToTrig`$\left[a\, e^{i t \omega}\left(\frac{1}{C} - i R \omega - L \omega^2\right)\right]$ `// Expand`

Out[110]= $\frac{a \operatorname{Cos}[t\,\omega]}{C} - i\, a R \omega \operatorname{Cos}[t\,\omega] - a L \omega^2 \operatorname{Cos}[t\,\omega] + \frac{i\, a \operatorname{Sin}[t\,\omega]}{C} + a R \omega \operatorname{Sin}[t\,\omega] - i\, a L \omega^2 \operatorname{Sin}[t\,\omega]$

We then take combine the numerator and denominator. Since the denominator is now real, this simply means take the real part of the numerator and keep the denominator unchanged.

In[111]:= `yprovisional = FullSimplify`$\left[\frac{\texttt{Re[\%]}}{\texttt{\%\%}}\right]$

Out[111]= $\frac{a C \left(\left(1 - C L \omega^2\right) \operatorname{Cos}[t\,\omega] + C R \omega \operatorname{Sin}[t\,\omega]\right)}{1 + C \omega^2 \left(C R^2 + L\left(-2 + C L \omega^2\right)\right)}$

Notice that we could have gotten to $y_{\text{provisional}}$ in one step using **ComplexExpand[]**:

In[112]:= 
```
yprovisional = ComplexExpand[Re[z[t]]]
```

Out[112]= $\frac{a\,\mathrm{Cos}[t\,\omega]}{C\left(R^2\omega^2+\left(\frac{1}{C}-L\omega^2\right)^2\right)}-\frac{a\,L\,\omega^2\,\mathrm{Cos}[t\,\omega]}{R^2\omega^2+\left(\frac{1}{C}-L\omega^2\right)^2}+\frac{a\,R\,\omega\,\mathrm{Sin}[t\,\omega]}{R^2\omega^2+\left(\frac{1}{C}-L\omega^2\right)^2}$

Also, here is the imaginary part. It looks similar, but the roles of sine and cosine have been reversed:

In[113]:= 
```
ComplexExpand[Im[z[t]]]
```

Out[113]= $-\frac{a\,R\,\omega\,\mathrm{Cos}[t\,\omega]}{R^2\omega^2+\left(\frac{1}{C}-L\omega^2\right)^2}+\frac{a\,\mathrm{Sin}[t\,\omega]}{C\left(R^2\omega^2+\left(\frac{1}{C}-L\omega^2\right)^2\right)}-\frac{a\,L\,\omega^2\,\mathrm{Sin}[t\,\omega]}{R^2\omega^2+\left(\frac{1}{C}-L\omega^2\right)^2}$

Let $\theta = \arctan\left(\frac{1}{C} - L\,\omega^2,\ -R\,\omega\right)$.

In[114]:= $\varphi = \mathrm{ArcTan}\left[\frac{1}{C} - L\,\omega^2,\ -R\,\omega\right]$

Out[114]= $\mathrm{ArcTan}\left[\frac{1}{C} - L\,\omega^2,\ -R\,\omega\right]$

- The two-variable arctangent gives an angle between $-\pi$ and $\pi$, taking into account the quadrant of the point.

Then the expression in the **Evaluate[]** in the cell below is equivalent to $y_{\text{provisional}}$. Notice that they have the same denominator, and the equality of the numerators follows from the angle addition formulas and then basic facts $\cos(\arctan(x, y)) = x$, $\sin(\arctan(x, y)) = y$. Just to be sure, we ask *Mathematica* to verify:

In[115]:= $\mathrm{FullSimplify}\left[\mathrm{Evaluate}\left[\frac{a\,\mathrm{Cos}[\omega\,t+\varphi]}{\left(\frac{1}{C}-L\,\omega^2\right)^2+(R\,\omega)^2}\ //\ \mathrm{TrigExpand}\right] == \mathrm{yprovisional},\right.$
```
  TransformationFunctions →
   {Automatic, (# /. {Cos[ArcTan[d_, n_]] -> d, Sin[ArcTan[d_, n_]] -> n}) &}]
```

Out[115]= True

So, the general solution is one of the following  The only remaining issue is what is the size of $4\,L$ versus $C\,R^2$, which affects the homogeneous solution. When $4\,L$ is smaller, we have two decaying exponentials. When $4\,L$ is larger, we have two oscillating, decaying exponentials. When they are equal, we have an exponential and a polynomial exponential.  This is the electrical equivalent of a critically damped harmonic oscillator, in which the system returns to equilibrium as quickly as possible. Clearly it is a very interesting case from an engineering standpoint. So here are the three possible solutions:

In[116]:= $\mathrm{yOverDamped}[t\_] := \frac{a\,\mathrm{Cos}[\omega\,t+\varphi]}{\left(\frac{1}{C}-L\,\omega^2\right)^2+(R\,\omega)^2} + c1\,e^{\frac{-R+\sqrt{-\frac{4L}{C}+R^2}}{2L}t} + c2\,e^{\frac{-R-\sqrt{-\frac{4L}{C}+R^2}}{2L}t}$

$\mathrm{yCriticallyDamped}[t\_] := \frac{a\,\mathrm{Cos}[\omega\,t+\varphi]}{\left(\frac{1}{C}-L\,\omega^2\right)^2+(R\,\omega)^2} + c1\,e^{\left(\frac{-C\,R}{2L}\right)t} + c2\,t\,e^{\left(\frac{-C\,R}{2L}\right)t}$

$\mathrm{yUnderDamped}[t\_] :=$

$\frac{a\,\mathrm{Cos}[\omega\,t+\varphi]}{\left(\frac{1}{C}-L\,\omega^2\right)^2+(R\,\omega)^2} + c1\,e^{\left(\frac{-C\,R}{2L}\right)t}\,\mathrm{Cos}\left[\frac{\sqrt{-\frac{4L}{C}+R^2}}{2L}\,t\right] + c2\,e^{\left(\frac{-C\,R}{2L}\right)t}\,\mathrm{Sin}\left[\frac{\sqrt{-\frac{4L}{C}+R^2}}{2L}\,t\right]$

All in three cases, however, the homogeneous solution decays to zero, leaving just the driven solution asymptotically. This piece has its largest amplitude when the denominator is minimized, which for fixed $R$ occurs at $\frac{1}{C} = L\,\omega^2$, or $\omega = \sqrt{\frac{1}{LC}}$. Recall that $\sqrt{\frac{1}{LC}}$ is what we found for the natural frequency of an $LC$ circuit, so the amplitude is maximized at resonance. Indeed, if there is no resistance, the amplitude will grow without bound, as we found on problem 3.4.21.

Being finished with our assumptions, we reset them by assigning **True** to **$Assumptions**

```
In[119]:= $Assumptions = True

Out[119]= True
```

# Is that a parameter which I see varying before me?

### Who said we need constant coefficients?

In[120]:=
```
Clear[PofD, y]
PofD[t_] := DPlusS[t]^2 - 2/t DPlusS[t] + 2/t^2 DPlusS[t]^0
```

### Basic solution set

We can construct a basic solution set in the standard manner

In[122]:=
```
DSolve[{PofD[t][y[t]] == 0, y[1] == 1, y'[1] == 0}, y[t], t]
DSolve[{PofD[t][y[t]] == 0, y[1] == 0, y'[1] == 1}, y[t], t]
Wronskian[y[t] /. Join[%, %%], t]
```

Out[122]= $\{\{y[t] \to 2t - t^2\}\}$

Out[123]= $\{\{y[t] \to -t + t^2\}\}$

Out[124]= $-t^2$

By adding and subtracting the first to/from the second, we would get $\{t, t^2\}$ is a basic solution set

In[125]:=
```
Clear[y1, y2]
y1[t_] := t
y2[t_] := t^2
PofD[t][y1[t]]
PofD[t][y2[t]]
Wronskian[{y1[t], y2[t]}, t]
```

Out[128]= 0

Out[129]= 0

Out[130]= $t^2$

### Variation of Parameters

Consider a logarithmic forcing function

In[131]:=
```
f[t_] := Log[t]^2
```

We can construct the

In[132]:=
```
yGen[t_] = Simplify[Integrate[-y2[s] f[s] / Wronskian[{y1[s], y2[s]}, s], {s, 1, t}] y1[t] +
    Integrate[y1[s] f[s] / Wronskian[{y1[s], y2[s]}, s], {s, 1, t}] y2[t] +
    c1 y1[t] + c2 y2[t], Assumptions → t > 0] // Expand
```

Out[132]= $2\,t + c1\,t - 2\,t^2 + c2\,t^2 + 2\,t^2\,\mathrm{Log}[t] - t^2\,\mathrm{Log}[t]^2 + \frac{1}{3}\,t^2\,\mathrm{Log}[t]^3$

In[133]:=
```
PofD[t][yGen[t]] // Simplify
yGen[1]
yGen'[1]
```

Out[133]= $\mathrm{Log}[t]^2$

Out[134]= $c1 + c2$

Out[135]= $c1 + 2\,c2$

In[136]:=
```
Plot[Evaluate[yGen[t] /. {c1 → 0, c2 → 0}], {t, 0, 2}]
```

Out[136]=


## Use and Reception

- Code given as black box to students
  - advanced CS students can analyze/appreciate it
- Students assigned these problems, including higher order equations
- Students had little difficulty mastering the use these operators and showed they could self correct if started with wrong polynomial
- Emphasize operator/linear algebra view
  - There is a single theory of linear equations!